Group Explorer Assistant!!!

By: **Ken Monks**
Professor of Mathematics
Dept. of Mathematics
University of Scranton
Scranton, PA 18510
monks@uofs.edu

This Maple worksheet contains a utility that converts Maple 3d plot objects to the .gp file format used by the excellent program GroupExplorer by Nathan Carter. You may use it freely as long as you give credit to the original author and indicate clearly any modifications you make to the document. I do not make any claims as to the accuracy or robustness of any of these routines, so you are on your own. Suggestions for improvement are always welcome. In order to use the routines in this worksheet, you must press ENTER on one of the input lines in the Source code section.

Source Code

In order to use the routines in this worksheet, you must press ENTER on one of the input lines in this section.

```
> restart;
  with(plots):with(plottools):

  MakeGEObject:=proc(
     ObjectNumber::posint,            # the Object sequence number
     Name::string,                    # the name of the object
     P::specfunc(anything,PLOT3D),    # a Maple 3d plot structure
     Operations::string)              # any operations are currently added by hand

     local
  nodecnt,linecnt,A,Nodes,Lines,text,thisColor,black,i,j,k,C,N,nodesize,thisCoord;

     # defaulty node size is 2 and default color for things is black
```

```
nodesize:=2; nodecnt:=0; linecnt:=0; A:=[op(P)];
Nodes:=select(x->type(x,specfunc(anything,POINTS)),A);
Lines:=select(x->type(x,specfunc(anything,CURVES)),A);
text:=cat("[Object",ObjectNumber,"]\nName = ",Name,"\n");
# process the nodes
for i in Nodes do

  # get the colors for the nodes
  thisColor:=GetColor(i):

  # get the coordinates
  N:=select(x->type(x,[numeric,numeric,numeric]),[op(i)]);
  for j in N do
     nodecnt:=nodecnt+1;
     thisCoord:=sprintf("<%08.5f,%08.5f,%08.5f> ",j[1],j[2],j[3]);
     text:=cat(text,"Node",nodecnt," = ",thisCoord,thisColor,nodesize,"\n");
  od:
od:

# process the lines
for i in Lines do

  # get the colors for the lines
  thisColor:=GetColor(i);

  # get the coordinates
  N:=select(x->type(x,list([numeric,numeric,numeric])),[op(i)]);
  for j in N do
     if nops(j)>1 then
       for k to nops(j)-1 do
         linecnt:=linecnt+1;
```

```
              thisCoord:=sprintf("<%08.5f,%08.5f,%08.5f> <%08.5f,%08.5f,%08.5f> "

  ,j[k][1],j[k][2],j[k][3],j[k+1][1],j[k+1][2],j[k+1][3]);
              text:=cat(text,"Line",linecnt," = ",thisCoord,thisColor,"0.0 N\n");
            od:
          fi:
        od:
     od:

     # check for Operations
     if nargs>3 then text:=cat(text,Operations); fi:

     text;
  end:
>
  # this utility is called by MakeGEObject to strip the colors
  # from a Maple plot object primitive
  GetColor:=proc(i)
     local black, thisColor, C;

     black:="<0.0,0.0,0.0> ";
     thisColor:=black:
     C:=select(x->type(x,Or(COLOUR(identical(RGB),numeric,numeric,numeric),
                            COLOR(identical(RGB),numeric,numeric,numeric),
                            COLOUR(identical(HUE),numeric),
                            COLOR(identical(HUE),numeric)
                           )
                       ),
               [op(i)]
              );
     if nops(C)>0 then
```

```
      C:=op(1,C):
      if op(1,C)=RGB then
         thisColor:=sprintf("<%04.2f,%04.2f,%04.2f> ",op(2,C),op(3,C),op(4,C));
      elif op(1,C)=HUE then
         thisColor:=sprintf("Hue%04.2f ",op(2,C));
      fi:
   fi:
   thisColor;
end:

# Converts a numeric HUE value to RGB
# based on code by Nathan Carter
HueToColor:=proc(f::numeric)
  local r, g, b, x, hue;

  hue := frac(f);
  x := trunc(256 * frac(hue * 6));
  if hue < 1/6 then
    r := 255;  g := x;  b := 0;
  elif hue < 2/6 then
    r := 255 - x;  g := 255;  b := 0;
  elif hue < 3/6 then
    r := 0;  g := 255;  b := x;
  elif hue < 4/6 then
    r := 0;  g := 255 - x;  b := 255;
  elif hue < 5/6 then
    r := x;  g := 0;  b := 255;
  else
    r := 255;  g := 0;  b := 255 - x;
  end;
  r/256.,g/256.,b/256.;
```

```
  end:

  # This makes a pinwheel as a symmetry object for the cyclic group of order n:
  Pinwheel:=proc(n)
     local OneCurve,ManyCurves,Node,i;

     OneCurve := spacecurve([t,0,-sin(Pi*t)/(1.75+n/4)],t=0..1,
                 color=black,numpoints=10):
     ManyCurves := NULL:
     for i from 0 to n-1 do
       Node:=rotate(point([1,0,0],color=COLOR(HUE,evalf(1-i/n))),0,Pi*2*i/n,0);
       ManyCurves := ManyCurves,rotate(OneCurve,0,Pi*2*i/n,0),Node:
     od:
     rotate(display(ManyCurves,symbol=circle),0,0,Pi/2);
  end:

  # Cut and paste doesn't work very well, so we save to a text file
  SaveStringToFile:=proc(S::string,fname::string)
    local fid;
    try
      fid:=fopen(fname,WRITE,TEXT):
    catch :
     fclose(fid);
     error "Unable to open file";
    end:
    fprintf(fid,S):
    fclose(fid);
    "OK";
  end:
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

Documentation

MakeGEObject()

Calling Sequence:

```
MakeGEObject(ObjectNumber, Name, Plot, Operations)
```

Parameters

```
ObjectNumber  - a positive integer representing the Object number in .gp files
                e.g. if ObjectNumber=3 the the first line in the output will be:
                [Object3]
Name          - a string containing the value of the .gp Name field
Plot          - any Maple PLOT3D object
Operations    - (optional) a string containing the .gp file Operations lines
```

Description:

- `MakeGEObject()` converts any Maple 3d plot object into a single string that can be saved to a text file for inclusion into a GroupExplorer .gp definition file.
- It converts any POINTS and CURVES in the Maple PLOT3D object into Nodes and Lines of the same color in the .gp file. All other plot information contained in the Maple plot object is simply ignored.
- Since there are problems with using cut and paste to copy the string from the Maple worksheet to a .gp file, you can save the string produced by `MakeGEObject()` to a text file by using the provided utility `SaveStringToFile()`.
- The syntax for `SaveStringToFile` is `SaveStringToFile(S,Fname)` where `S` is the string to save and `Fname` is the full pathname of the file you want to save it to. Note that when specifiying filenames in Maple strings you must double the backslashes so that `"C:\math\MyFile.txt"` has to be written as `"C:\\math\\MyFile.txt"`.

Examples

We can use any Maple commands whatsoever to create our 3d plots. Some useful Maple packages for this purpose are plottools, geom3d,

and plots.

A Simple Example

In this example we will use the plottools and plots packages.

```
> with(plots):with(plottools):
```

Define some 3d points for nodes:

```
> A:=[-2,2,0]:
  B:=[2,2,0]:
  C:=[0,-2,0]:
```

Now it is trivial to make the 3d plot object using the Maple tools. In this case we make a figure containing the three points, a line segment connecting two of them, and use the spacecurve command to plot a parabola.

```
> P:=display(point(A,color=red),
             point(B,color=green),
             point(C,color=blue),
             line(A,B,color=black),
             spacecurve([t,t^2-2,0],t=-2..2,color=black)):
```

Now here is a fun part. The plot we produced above is too big for the viewing window in GroupExplorer, so we just scale the whole thing by a factor of 0.5. Then we display it in Maple to preview it.

```
> P:=scale(P,0.5,0.5,0.5,[0,0,0]):
  display(rotate(P,Pi/2,0,0),scaling=constrained,axes=framed,symbol=circle,view=[-1.1..1
  .1,-1.1..1.1,-1.1..1.1]);
```


1
0.5
0
-0.5
-1
-1
-0.5
0
0.5
1
1
0.5
0
-0.5
-1

We are now ready to convert our plot object to a string (change the : to a ; on the next line if you want to see the string before saving it).

```
> S:=MakeGEObject(1,"Wacky Parabola Thing",P):
```

And save it to disk (replace the filename with your own filename).

```
> SaveStringToFile(S,"C:\\math\\GroupExplorer\\mapleout.txt");
```

"OK"

A More Advanced Example

The following example is by Nathan Carter.
It produces an object of symmetry for the cyclic group of order n by constructing one piece of the object and then using Maple to rotate that piece to construct the rest of the object.

Put order of your cyclic group here:

```
> n := 8:
```

Get one little bit of the object of symmetry:

```
> OneCurve := display([
    point([1,0,0],color=red),
    spacecurve([t,0,-sin(Pi*t)/(1.75+n/4)],t=0..1,
      color=black,numpoints=10)
    ]):
```

Make many of them, in a fan shape:

```
> ManyCurves := NULL:
  for i from 0 to n-1 do
    ManyCurves := ManyCurves,rotate(OneCurve,0,Pi*2*i/n,0):
  end do:
```

Turn it into a plot object and preview it:

```
> Q := display(ManyCurves,
                scaling=constrained,axes=framed,symbol=circle):
  rotate(Q,Pi/2,0,0);  # so I can preview how it will look in GE
```


1
0.5
0
-0.5
-1
-1
-0.5
0
0.5
1
1
0.5
0
-0.5
-1

Looks good! Save it to a file (replace the file name with your own file name):

```
> SaveStringToFile(
    MakeGEObject(1,sprintf("Propeller with %d blades",n),Q),
    "C:\\math\\GroupExplorer\\mapleout.txt"
    );
```

"OK"

The Pinwheel command automates this construction and gives different colored nodes:

```
> Pinwheel(16);
```



A really advanced example

In this example we generate the entire .gp file for Z_n for n from 2 to 20:

```
> for n from 2 to 20 do
    SaveStringToFile(
      cat("[Group]\n",
        "Name1 = Z_",n,"\n",
        "Name2 = Cyclic group of order ",n,"\n",
        "Author = Nathan Carter & Ken Monks\n",
        "Generators1 = ",seq(cat(i," "),i=1..n-1),"0\n",
        "Representation1 = Generators(e, a=",seq(cat(i," "),i=1..n-1),"0)\n",
        "Representation2 = Cycles\n\n",
        MakeGEObject(1,sprintf("Propeller with %d blades: front view",n),
                     Pinwheel(n),
                     cat("Operation1 = ",seq(cat(i," "),i=1..n-1),"0 : ",
                         round(360/n)," <1,0,0>\n\n")
                    ),
        MakeGEObject(2,sprintf("Propeller with %d blades: above view",n),
                     translate(rotate(Pinwheel(n),0,0,-Pi/2),0,-0.75,0),
                     cat("Operation1 = ",seq(cat(i," "),i=1..n-1),"0 : ",
                          round(360/n)," <0,1,0>\n\n")
                    )
      ),
      cat("C:\\math\\GroupExplorer\\Z_",n,".gp")
    );
  od:
```

Revision history

- Version 0.2.1 - July 17, 2003
  Fixed minor bugs in the really advanced example

- Version 0.2 - July 12, 2003
  Modified the MakeGEObject proc so it recognizes Maple colors specified by HUE in a plot object.
  Make GetColor() a separate proc since it is now mor complicated and has to be called for both Nodes and Lines.
- Version 0.1 - July 10, 2003
  Added the Pinwheel proc and the really advanced example
- Version 0.0.1 - July 4, 2003
  Added Nathan Carter's cyclic group object example
- Version 0.0 - June 25, 2003